

Module 20.3: nag_chisq_dist

Probabilities and Deviate for a χ^2 -distribution

`nag_chisq_dist` provides procedures for computing probabilities and the deviate for a χ^2 -distribution.

Contents

Procedures

<code>nag_chisq_prob</code>	20.3.3
Computes lower or upper tail probability for a χ^2 -distribution with ν degrees of freedom	
<code>nag_chisq_deviate</code>	20.3.5
Computes the deviate associated with a given lower tail probability of a χ^2 -distribution with ν degrees of freedom	

Examples

Example 1: Calculation of probabilities and the deviate for a χ^2 -distribution.....	20.3.7
---	--------

Additional Examples	20.3.9
----------------------------------	--------

References	20.3.10
-------------------------	---------

Procedure: nag_chisq_prob

1 Description

`nag_chisq_prob` returns the lower or upper tail probability for a χ^2 -distribution with ν degrees of freedom.

2 Usage

USE `nag_chisq_dist`

[*value* =] `nag_chisq_prob`(*tail*, *x*, *df*, [, *optional arguments*])

The function result is a scalar of type `real(kind=wp)`.

3 Arguments

3.1 Mandatory Arguments

tail — `character(len=1)`, `intent(in)`

Input: the type of tail probability to be returned:

if **tail** = 'L' or 'l', the lower tail probability is returned;

if **tail** = 'U' or 'u', the upper tail probability is returned.

Constraints: **tail** = 'L', 'l', 'U' or 'u'.

x — `real(kind=wp)`, `intent(in)`

Input: the value of the χ^2 variate.

Constraints: **x** $\geq$ 0.0.

df — `real(kind=wp)`, `intent(in)`

Input: the degrees of freedom, ν , of the χ^2 -distribution.

Constraints: **df** $>$ 0.0.

3.2 Optional Argument

error — `type(nag_error)`, `intent(inout)`, optional

The NAG *f*90 error-handling argument. See the Essential Introduction, or the module document `nag_error_handling` (1.2). You are recommended to omit this argument if you are unsure how to use it. If this argument is supplied, it *must* be initialized by a call to `nag_set_error` before this procedure is called.

4 Error Codes

Fatal errors (`error%level = 3`):

<code>error%code</code>	Description
301	An input argument has an invalid value.

Warnings (error%level = 1):

error%code	Description
101	The solution has failed to converge. The result is, however, a reasonable approximation.

5 Examples of Usage

A complete example of the use of this procedure appears in Example 1 of this module document.

6 Further Comments

6.1 Mathematical Background

Given that x is a χ^2 distributed variate with ν degrees of freedom, the lower tail probability $P(X \leq x : \nu)$ is defined by:

$$P(X \leq x : \nu) = \frac{1}{2^{\nu/2}\Gamma(\nu/2)} \int_{0.0}^x X^{\nu/2-1} e^{-X/2} dX, \quad x \geq 0, \quad \nu > 0.$$

6.2 Algorithmic Detail

To calculate $P(X \leq x : \nu)$, a transformation of a gamma distribution is employed: i.e., a χ^2 -distribution with ν degrees of freedom is equal to a gamma distribution with scale parameter 2 and shape parameter $\nu/2$.

6.3 Accuracy

In most cases the result is relatively accurate to 5 significant figures.

Procedure: nag_chisq_deviate

1 Description

`nag_chisq_deviate` returns the deviate associated with the lower tail probability of a χ^2 -distribution with ν degrees of freedom.

2 Usage

USE `nag_chisq_dist`

[*value* =] `nag_chisq_deviate`(*p*, *df* [, *optional arguments*])

The function result is a scalar of type `real(kind=wp)`.

3 Arguments

3.1 Mandatory Arguments

p — `real(kind=wp)`, `intent(in)`

Input: the lower tail probability of the χ^2 -distribution.

Constraints: $0.0 \leq p < 1.0$.

df — `real(kind=wp)`, `intent(in)`

Input: the degrees of freedom, ν , of the χ^2 -distribution.

Constraints: `df` > 0.0.

3.2 Optional Argument

error — `type(nag_error)`, `intent(inout)`, optional

The NAG *f*90 error-handling argument. See the Essential Introduction, or the module document `nag_error_handling` (1.2). You are recommended to omit this argument if you are unsure how to use it. If this argument is supplied, it *must* be initialized by a call to `nag_set_error` before this procedure is called.

4 Error Codes

Fatal errors (`error%level = 3`):

<code>error%code</code>	Description
301	An input argument has an invalid value.

Failures (`error%level = 2`):

<code>error%code</code>	Description
201	<i>p</i> is too close to 0.0 or 1.0 for the result to be calculated. You may try another value for <i>p</i>.
202	The series used for calculating the underlying solution has failed to converge. This error is not very likely to occur.

Warnings (error%level = 1):

error%code	Description
101	The solution has failed to converge. But the result is a reasonable approximation to the solution.

5 Examples of Usage

A complete example of the use of this procedure appears in Example 1 of this module document.

6 Further Comments**6.1 Mathematical Background**

Given that p is the lower tail probability of a χ^2 -distribution with ν degrees of freedom, the deviate x_p associated with p is defined as the solution to

$$P(X \leq x_p : \nu) = p = \frac{1}{2^{\nu/2}\Gamma(\nu/2)} \int_0^{x_p} e^{-X/2} X^{\nu/2-1} dX, \quad 0 \leq x_p < \infty, \quad \nu > 0.$$

6.2 Algorithmic Detail

The required x_p is found by using the relationship between a χ^2 -distribution and a gamma distribution: i.e., a χ^2 -distribution with ν degrees of freedom is equal to a gamma distribution with scale parameter 2 and shape parameter $\nu/2$.

For very large value of ν ($> 10^5$) Wilson and Hilferty's normal approximation to a χ^2 is used; see Kendall and Stuart [1].

6.3 Accuracy

Although the result is accurate to 5 significant digits for most parameter values, it may be poor if p is close to 0.0.

Example 1: Calculation of probabilities and the deviate for a χ^2 -distribution

This example program shows how `nag_chisq_prob` returns the lower tail probability or upper tail probability for a χ^2 -distribution with ν degrees of freedom. It also illustrates how `nag_chisq_deviate` calculates the deviate (`x_deviate`) associated with a given tail probability.

1 Program Text

Note. The listing of the example program presented below is double precision. Single precision users are referred to Section 5.2 of the Essential Introduction for further information.

```
PROGRAM nag_chisq_dist_ex01

! Example Program Text for nag_chisq_dist
! NAG f190, Release 3. NAG Copyright 1997.

! .. Use Statements ..
USE nag_examples_io, ONLY : nag_std_out, nag_std_in
USE nag_chisq_dist, ONLY : nag_chisq_prob, nag_chisq_deviate
! .. Implicit None Statement ..
IMPLICIT NONE
! .. Intrinsic Functions ..
INTRINSIC KIND
! .. Parameters ..
INTEGER, PARAMETER :: wp = KIND(1.0D0)
! .. Local Scalars ..
REAL (wp) :: df, prob, probl, x, x_calculated
CHARACTER (1) :: tail
! .. Executable Statements ..

WRITE (nag_std_out,*) 'Example Program Results for nag_chisq_dist_ex01'

READ (nag_std_in,*)          ! Skip heading in data file

WRITE (nag_std_out,*)
WRITE (nag_std_out,*) 'TAIL      X      DF      PROB      DEVIATE'
WRITE (nag_std_out,*)

DO
  READ (nag_std_in,*,end=20) tail, x, df

  prob = nag_chisq_prob(tail,x,df)

  probl = prob
  IF (tail=='u' .OR. tail=='U') probl = 1.0_wp - prob

  x_calculated = nag_chisq_deviate(probl,df)

  WRITE (nag_std_out,'(2X,A1,4x,F6.3,F8.3,F8.4,f10.4)') tail, x, df, &
    prob, x_calculated
END DO
20  CONTINUE

END PROGRAM nag_chisq_dist_ex01
```

2 Program Data

Example Program Data for nag_chisq_dist_ex01

```
'L'  8.26  20.0      :tail, x, df
'U'   6.2   7.5
'L'  55.76  45.0
```

3 Program Results

Example Program Results for nag_chisq_dist_ex01

TAIL	X	DF	PROB	DEVIATE
L	8.260	20.000	0.0100	8.2600
U	6.200	7.500	0.5721	6.2000
L	55.760	45.000	0.8694	55.7600

Additional Examples

Not all example programs supplied with NAG *f*/90 appear in full in this module document. The following additional examples, associated with this module, are available.

`nag_chisq_dist_ex02`

Calculation of the deviate associated with a given lower tail probability for a χ^2 -distribution with known degrees of freedom.

References

- [1] Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* Griffin (3rd Edition)